AN INTRODUCTION TO MAPLE FOR ECONOMICS 531

This file provides an introduction to some basic functions in Maple

1. Plotting in Two Dimensions

```
> restart:
> with(plots):
```

Warning, the name changecoords has been redefined

1.1 Basic Plot

```
> f:=a*x^2+b*x-c;
```

$$f := a\,x^2 + b\,x - c$$

```
> f1:=subs(a=1,b=2,c=3,f);
```

$$f1 := x^2 + 2\,x - 3$$

```
> plot(f1,x=-5..5);
```



1.2 Multiple Plot

```
> f2:=subs(a=1,b=2,c=0,f);
```

$$f2 := x^2 + 2\,x$$

```
> g1:=plot(f1,x=-5..5,color=red):
```

```
> g2:=plot(f2,x=-5..5,color=blue):
> display(g1,g2);
```



1.3 Labeling the Plot

```
> g3:=textplot([4,16,'f1']):
> g4:=textplot([4,28,'f2']):
> display(g1,g2,g3,g4);
```



1.4 Implicit Plots

A Constrained Optimization Example

Cobb-Douglas Utility

```
> u:=x1^a*x2^b;
```

$$u := x1^a\ x2^b$$

```
> a:=1/2:
> b:=1/2:
> p1:=1:
> p2:=1:
> m:=10:
```

```
> g1:=implicitplot(u=2,x1=0..10,x2=0..10):
> g2:=implicitplot(u=3.5,x1=0..10,x2=0..10):
> g3:=implicitplot(u=5,x1=0..10,x2=0..10):
> g4:=implicitplot(u=6,x1=0..10,x2=0..10):
> g5:=implicitplot(p1*x1+p2*x2=m,x1=0..10,x2=0..10,colour=blue):
> display(g1,g2,g3,g4,g5);
```



Resetting parameter values:

```
> a:='a':
> b:='b':
```

1.5 Plotting a Constant

```
> g1:=implicitplot(x=5,x=0..10,y=0..10):
> display(g1);
```




```
> g1:=implicitplot(y=5,x=0..10,y=0..10):
> display(g1);
```



2. Plotting in Three Dimensions

```
> u;
```

$$x1^a\, x2^b$$

```
> a:=1/2:
> b:=1/2:
>
> g1:=plot3d(u,x1=0..10,x2=0..10,axes=boxed):
> g2:=implicitplot3d(p1*x1+p2*x2=m,x1=0..10,x2=0..10,y=0..10,axes=bo
xed,colour=blue):
> display(g1,g2);
```



3. Solving Equations

```
> restart:
```

3.1 A Single Equation

```
> f:=a*x^2+b*x-c;
```

$$f := a\,x^2 + b\,x - c$$

```
> eqn:=f=0:
> soln:=solve(eqn,x);
```

$$soln := \frac{-b + \sqrt{b^2 + 4\,a\,c}}{2\,a}, \frac{-b - \sqrt{b^2 + 4\,a\,c}}{2\,a}$$

```
> soln1:=soln[1];
```

$$soln1 := \frac{-b + \sqrt{b^2 + 4\,a\,c}}{2\,a}$$

```
> soln2:=soln[2];
```

$$soln2 := \frac{-b - \sqrt{b^2 + 4\,a\,c}}{2\,a}$$

3.2 A System of Equations

A Constrained Optimization Example

```
> u:=x1^a*x2^b;
```

$$u := x1^a\, x2^b$$

```
> L:=u+lambda*(m-p1*x1-p2*x2);
```

$$L := x1^a\, x2^b + \lambda\,(m - p1\, x1 - p2\, x2)$$

```
> eqn1:=diff(L,x1)=0;
```

$$eqn1 := \frac{x1^a\, a\, x2^b}{x1} - \lambda\, p1 = 0$$

```
> eqn2:=diff(L,x2)=0;
```

$$eqn2 := \frac{x1^a\, x2^b\, b}{x2} - \lambda\, p2 = 0$$

```
> eqn3:=diff(L,lambda)=0;
```

$$eqn3 := m - p1\, x1 - p2\, x2 = 0$$

```
> soln:=solve({eqn1,eqn2,eqn3},{x1,x2,lambda});
```

$$soln := \left\{ \lambda = \frac{\mathbf{e}^{\left(\ln\left(\frac{m\, b}{p2\,(a+b)}\right) b + \ln\left(\frac{m\, a}{(a+b)\, p1}\right) a\right)} (a+b)}{m},\; x1 = \frac{m\, a}{(a+b)\, p1},\; x2 = \frac{m\, b}{p2\,(a+b)} \right\}$$

```
> x1hat:=subs(soln,x1);
```

$$x1hat := \frac{m\, a}{(a+b)\, p1}$$

```
> x2hat:=subs(soln,x2);
```

$$x2hat := \frac{m\, b}{p2\,(a+b)}$$

```
> lambdahat:=simplify(subs(soln,lambda));
```

$$lambdahat := \frac{\left(\frac{m\, b}{p2\,(a+b)}\right)^b \left(\frac{m\, a}{(a+b)\, p1}\right)^a (a+b)}{m}$$

The Maximum Value Function (The Indirect Utility Fuction)

```
> v:=subs(x1=x1hat,x2=x2hat,u);
```

$$v := \left(\frac{m\, b}{p2\,(a+b)}\right)^b \left(\frac{m\, a}{(a+b)\, p1}\right)^a$$

The Lagrange Mutliplier Again:

```
> simplify(diff(v,m));
```

$$\frac{\left(\frac{m\,b}{p2\,(a+b)}\right)^{b}\left(\frac{m\,a}{(a+b)\,p1}\right)^{a}(a+b)}{m}$$

4. Integration

```
> F:=int(x2hat,p2);
```

$$F := \frac{m\,b\,\ln(p2)}{a+b}$$

```
> F:=int(x2hat,p2=1..2);
```

$$F := \frac{\ln(2)\,m\,b}{a+b}$$

```
>
```